
第5回　アキュムレータマシン

前回、pcを取り付けて、命令を一つづつ取り出して実行する機構を作った。これでプログラム格納型コンピュータの

基本的な部分ができた。このコンピュータは、アキュムレータのみを持つ最も簡単なものでありアキュムレータマシ

ンと呼ばれる。

図 1: アキュムレータマシンの構成

図 1にアキュムレータマシンの構造を示す。この動作を復習しよう。まず、Instruction Memoryには pcがアドレ

スとして繋がっており、これによって指し示されたアドレスの命令が読み出さる。

読み出された命令が ST命令以外ならば、Data memoryから命令中の下位 8ビットをアドレスとしてデータが読み

出される。つまり LD 1ならば 1がアドレスに載る。そして読み出されたデータは ddatain から aluの b入力に入る。

aluの s入力には、opcodeの下位 3ビットが入っている。LD命令の場合、opcodeは 0001なので、comは 001とな

り、THBが実行されて、読み出されたデータが、クロックの立ち上がリで、そのまま accumにセットされる。同時

に pcには pc+1がセットされ、次の番地の命令が指し示される。

読み出された命令が ADD 2ならば、opcodeは 0110なので、comには 110が与えられ、aluは a入力と b入力の

加算を行う。a入力には accumの値が常に入力され、b入力は 2番地から読み出したデータがやってくる。加算結果

は EX状態の終わりに accumに書き込まれる。

ちなみに、それぞれの命令で、accumに結果が設定されるのは、次のクロックの立ち上がりなので、この結果が利

用可能になる（シミュレータ上で観測できる）のは、次の命令を実行するクロックサイクルになる。

メモリへの書き込みを行う ST命令でのみ、データパスの動作はやや異なっている。ST 3を実行する場合を図 2に

示す。ST命令のアドレスは 00000011であるので、最上位が 1ならば ST命令であるという判断ができる。ここで、

Data memoryの入力ポートには常に accumの出力が dataoutを経由して接続されている。アドレスは他の命令同様、

operandの値が与えられるため、この場合は 3である。ここで weを Hにすることで、data memoryに accumの値

を書き込む。

前回の設計で、以下の命令が利用可能である。Xは番地の 2進数を示し、-は使われないので何でも良いこと（ドン

トケアと呼ぶ）を示す。

ニーモニック　　機械語　　　　　意味

NOP 　　　　　 0000 --------: No Operation: 何もしない

LD X 　　　　　 0001 XXXXXXXX: Load X: acc-m<- X番地中の値

図 2: ST命令の実行例

AND X 　 　　　 0010 XXXXXXXX: AND X: acc-m <- acc-m AND X番地中の値

OR X 　　　 0011 XXXXXXXX: OR X: acc-m <- acc-m OR X番地中の値

SL 　　　 0100 --------: Shift Left: acc-m <- acc-m<<1

SR 　　　 0101 --------: Shift Right: acc-m <- acc-m>>1

ADD X 　　　 0110 XXXXXXXX: Add X: acc-m <- acc-m + X番地中の値

SUB X 　　　 0111 XXXXXXXX: S-b X: acc-m <- acc-m - X番地中の値

ST X 　　　 1000 XXXXXXXX: Store X: accum -> X番地中

ここで、NOPはアキュムレータの値が自分自身に書き込まれるため、何もしない命令である。何もしない命令なん

て意味があるんだろうか？と思われるかもしれないが、実はこの命令は時間稼ぎのために重要で、全てのコンピュー

タで装備されている。

さて、このアキュムレータマシンは、命令メモリに格納された命令を順番に実行するに過ぎない。前回の課題のプ

ログラムとて、実行が終わったら、次の番地の命令の実行に移ってしまう。完全な猪突猛進型で、演算処理の終了後

に、停止することすらできない。判断を行い、実行する命令を変更する機能がなければ、計算アルゴリズムを実行す

ることができず、つまりコンピュータとは言えない。では、コンピュータとするためにはどうすれば良いだろう？

最も簡単な方法は、アキュムレータの内容を判断して、この結果によってプログラムカウンタを変更してしまうこ

とである。これが分岐 (Branch)命令である。ここでは以下の命令を導入する。

機械語　　　　命令

1001 XXXXXXXX BEZ X (Branch Equal Zero) X

1010 XXXXXXXX BNZ X (Branch Not equal Zero) X

BEZ命令は、アキュムレータの内容が 0ならば、PCの内容をXXXXXXXXに変更する。BNZ命令は、アキュム

レータの内容が 0でなければ、PCの内容を XXXXXXXXに変更する。条件に合わない場合は何もせず、次の命令が

実行される。

分岐命令の導入によって、アキュムレータの内容によって、動作する命令を変更して、反復計算をすることが可能

である。

以下がその例である。この例は、2番地の内容をm、3番地の内容を nとし、mを n回加算するプログラム、つま

り乗算のプログラムである。答えは 0番地に格納される。ただし、0番地を 0に、1番地を 1に初期化しておく必要

がある。アキュムレータが 0になるまで、6番地の分岐命令によって 0番地まで戻り、反復して実行する。その度に

3番地の内容である nが 1ずつ減っていき、0になると 7番地の内容が実行される。

番地　命令

0 　 LD 0

1 　 ADD 2

2 　 ST 0

3 　 LD 3

4 　 SUB 1

5 　 ST 3

6 　 BNZ 0

7 　 BEZ 7

この場合、反復が終了した時には、アキュムレータの内容が 0になっており、7番地の分岐命令は無限ループに陥

る。これをダイナミックストップと呼び、これによってプログラムは平和に停止する。

分岐命令を実行するためには、(1)命令メモリから読み出した命令が分岐命令であり、(2)条件が成立する場合に限

り、PCに命令の下位 8ビットの飛び先をセットできるようにすれば良い。

図 3: アキュムレータマシンの構成 (分岐命令付き)

図 refbranchにこの改造したアキュムレータマシンのハードウェア構成を示す。台形印は、マルチプレクサあるい

はデータセレクタというハードウェアモジュールで、この場合、2つの入力の片方を、選択信号 pcselに応じて選ん

で出力する。ここでは、pcsel=1ならば、pcに operandを送り、S=0ならば、pcには以前通り pc+1を送る。マル

チプレクサは、今後データパスを複雑にしていく場合不可欠なものである。また、accumには、accset信号を Lにし

て、分岐命令実行時にはデータをセットしないようにしてやる。この辺は ST命令と同様である。太線に BEZ, BNZ

実行時のデータの流れを示す。まとめると、命令コードと accumの値に応じて pcsel, accset, we, comの値を以下の

ように制御する必要がある。

• ALU演算命令 (LD,NOPを含む): accset=1, pcsel=0,we=0, com:opecodeの下位 3ビット

• ST命令:accset=0, pcsel=0,we=1,com:ドントケア

• 分岐命令：accset=0, pcsel:条件が成立すれば 1、そうでなければ 0、we=0、com:ドントケア

この信号はコントローラ (Cont)で生成される。コントローラはここでは非常に簡単な組み合わせ回路で済むが、命

令が増えるにつれて複雑になっていく。

イミーディエイト命令

今までの演算命令では、命令の下位 8ビットはメモリのアドレスであった。このため、例えば、1を引くという操

作を行う場合、先のプログラムで示したようにあらかじめ 1番地に 1という数字を初期化操作で入れておく必要があ

る。これでは不便であるため、コンピュータの命令セットにはイミーディエイトまたは直値（即値）という使い方が

用意されている。この方法では、命令コードの下位 8ビットはメモリのアドレスを示さず、そのまま数値として計算

に使われる。例えば以下の命令である。

機械語　　　　命令

1100 XXXXXXXX ADDI #X (ADD Immediate) X: accum <- accum + X

ADDIは、アキュムレータに対して下位 8ビットで示した数値を直接加算する命令である。これを使うと乗算のプ

ログラムは以下のようになる。この場合、1番地を 1に初期化する必要はない。このような命令を直値命令あるいは

即値命令と呼ぶが、どちらも発音しにくい上意味が取り難いせいか、英語読みでそのままイミーディエイト命令と呼

ぶ人も多い。

番地　命令

0 　 LD 0

1 　 ADD 2

2 　 ST 0

3 　 LD 3

4 　 ADDI #-1

5 　 ST 3

6 　 BNZ 0

7 　 BEZ 7

ちなみに、アセンブラ表現では、イミーディエイト命令であることを強調するために、計算に使われる直値を示す

部分に井桁印を付ける習慣がある。さて、命令の下位 8ビットは、メモリのアドレスを表す場合、分岐の飛び先を示す

場合、イミーディエイト命令の直接演算するデータを示す場合の三通りが考えられる。いずれも演算や処理の対象を

示すことから、この部分をオペランド（operand)と呼ぶ。すなわち、一般的に、命令コードは処理内容を示す opcode

とオペランドから構成される。

ここで一つ問題がある。今、アキュムレータに格納されているデータは 16ビットであるのに対して、命令中のオ

ペランドは 8ビットしかないので、データ幅が合わない。今、上記の例に従って、ADDI #-1の機械語は、-1を 2の

補数表現で表すため、1100 11111111となる。ところが、上位 8ビットを単純に 0で埋めると、0000000011111111と

なり、16ビットの数値としては+127となってしまう。すなわち、この場合、最上位ビットが 1ならば 1を、0なら

ば 0を上位 8ビットに入れて全体として 16ビットの値を作る必要がある。これを符号拡張 (sign extension)と呼ぶ。

8bitの-1を符号拡張すると、1111111111111111となり、16ビットでも-1として扱うことができる。

イミーディエイト命令を実装するには、ALUの B入力に対してマルチプレクサを付けてやり、メモリからのデー

タだけではなく、命令のオペランド部を入れてやることのできるようにする。次に、ALUの演算を ADDI命令実行

時に加算 (110)に設定してやる。これを付けた構成を図 4に示す。太線は、ADDI命令実行時のデータを動きを示す。

コントローラはADDI命令の場合に以下のように制御線に値を与えれば良い。accset=1, op addi=1, pcsel=0, we=0,

com=110(加算)。ちなみに新しいマルチプレクサの制御線である op addiは、ADDI命令の時だけ 1にする。

図 4: アキュムレータマシンの構成 (完成版)

Verilog記述

それでは、アキュムレータマシンを Verilogで記述してみよう。

今までは、ALU命令主体であったため、opcodeはその下位 3ビットをALUのコマンドとして使って、あとは ST

命令を識別するだけで良かった。しかし、命令が増えてくると、命令に応じてデータパスのデータの流れを制御する必

要がある。これが図中に示したコントローラである。コントローラは、まず opcodeの部分を解読（デコード）する。

wire [‘OPCODE_W-1:0] opcode;

wire [‘ADDR_W-1:0] operand;

wire op_st, op_bez, op_bnz, op_addi ;

assign idatain = imem[pc];

assign {opcode, operand} = idatain;

assign op_st = opcode == ‘OP_ST;

assign op_bez = opcode == ‘OP_BEZ;

assign op_bnz = opcode == ‘OP_BNZ;

assign op_addi = opcode == ‘OP_ADDI;

上記のコードでは、命令メモリから読んできた命令コード (idatain) の上位 4ビットを opcode、下位 4ビットを

operandとして分離する。そして opcodeを所定のビットパターンと比較して、それぞれの命令を識別する。

次に分岐命令を実装してみよう。分岐命令は、分岐命令実行時に条件が成立した時のみ、pcに飛び先の operandを

セットし、そうでなければ 1加算した結果をセットする。これは図中ではマルチプレクサで表されている。これを直

接 Verilogで記述すると以下のようになる。nextpcを生成する文がマルチプレクサに相当する。

wire pcsel, pcset;

wire [‘INST_W-1:0] nextpc;

assign pcsel = (op_bez & acc == 0 )| (op_bnz & acc != 0);

assign nextpc = pcsel ? operand : pc + 1;

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if (pcsel) pc <= nextpc;

else pc <= pc + 1;

end

この方法だと、マルチプレクサで次の pcを選んで、これを書き込む様子が明示的に表現されている。しかし、これは

中間的な信号である nextpcや pcselが必要で面倒であるともいえる。そこで、直接 if文中に条件を記述しても良い。

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if (op_bez & (accum == 0) | op_bnz & (accum != 0))

pc <= operand;

else pc <= pc + 1;

end

この方法は、マルチプレクサの機能を if文中の条件で表してしまっている。今回はこの方が見通しが良いだろう。さ

て、分岐命令の実行時には、ST命令と同様に、accumはALUの演算結果をセットしない。図で考えると accsetを 0

にする必要がある。しかし、これも、上の記述と同様に、わざわざ accsetを定義しないで、

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= 0;

else if(!op_st & !op_bez & !op_bnz ) accum <= alu_y;

end

と書けば良い。これで、STでもBEZでもBNZでもない命令では、aluで計算した値を accumにセットしてくれる。

このように、レジスタに対する設定は、if文が利用可能なので、条件をうまく書くと、マルチプレクサの存在を意

識しないで見通しの良い記述ができる。しかし、if文中の条件が複雑になりすぎると、かえって読み難いため、中間

的な信号線を定義した方が良い場合もある。この辺をどう書くかは、個人の趣味の範囲に入り、読みやすさを心がけ

つつ、好きなように書いて欲しい。

次に、イミーディエイト命令のADDIを実装しよう。この命令ではまず、オペランド部を符号拡張する必要がある。

これは、Verilogでは繰り返しと連結を利用して以下のように書く。

{{8{operand[7]}},operand}

この記述で operandの最上位ビットが 8個分 operand自体に連結される。次に ALUの b入力用のマルチプレクサ

を以下のように記述する。

wire [‘DATA_W-1:0] alu_b;

assign alu_b = op_addi ? {{8{operand[7]}},operand} : ddatain;

次に、ALUの演算を ADDI命令実行時に加算 (110)に設定する記述は以下の通りである。

wire [‘SEL_W-1:0] com;

assign com = op_addi ? ‘ALU_ADD : opcode[‘SEL_W-1:0];

alu alu_1(.a(accum), .b(alu_b), .s(com), .y(alu_y));

全体をまとめて以下に示す。

‘include "def.h"

module accum(

input clk, input rst_n,

output [‘DATA_W-1:0] accout);

reg [‘DATA_W-1:0] accum;

reg [‘ADDR_W-1:0] pc;

reg [‘DATA_W-1:0] dmem [‘DEPTH-1:0];

reg [‘INST_W-1:0] imem [‘DEPTH-1:0];

wire [‘DATA_W-1:0] alu_y;

wire [‘DATA_W-1:0] alu_b;

wire [‘DATA_W-1:0] ddatain;

wire [‘OPCODE_W-1:0] opcode;

wire [‘INST_W-1:0] idatain;

wire [‘ADDR_W-1:0] operand;

wire we;

wire op_st, op_bez, op_bnz, op_addi ;

wire [‘SEL_W-1:0] com;

assign idatain = imem[pc];

assign {opcode, operand} = idatain;

assign op_st = opcode == ‘OP_ST;

assign op_bez = opcode == ‘OP_BEZ;

assign op_bnz = opcode == ‘OP_BNZ;

assign op_addi = opcode == ‘OP_ADDI;

assign ddatain = dmem[operand];

assign we = op_st;

always @(posedge clk)

if(we) dmem[operand] <= accum;

assign accout = accum;

assign com = op_addi ? ‘ALU_ADD : opcode[‘SEL_W-1:0];

assign alu_b = op_addi ? {{8{operand[7]}},operand} : ddatain;

alu alu_1(.a(accum), .b(alu_b), .s(com), .y(alu_y));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if (op_bez & (accum == 0) | op_bnz & (accum != 0))

pc <= operand;

else pc <= pc + 1;

end

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= 0;

else if(!op_st & !op_bez & !op_bnz ) accum <= alu_y;

end

endmodule

シミュレーションに利用する。テストベンチは、前回とほとんど同じで良い。ただ、最後のシミュレーションの実

行クロック数 (finish文の前)のみを、場合に応じて変更する必要がある。例えば、

#STEP

rst_n <= ‘DISABLE_N;

#(STEP*50)

$finish;

とすれば、大方の場合は大丈夫かと思う。実装した accum.vで、掛け算のプログラムを実行してみよう。imem.dat

を下のように設定する。

0001_00000000 // LD 0

0110_00000010 // ADD 2

1000_00000000 // ST 0

0001_00000011 // LD 3

1100_11111111 // ADDI #-1

1000_00000011 // ST 3

1010_00000000 // BNZ 0

1001_00000111 // BEZ 7

このプログラムでは 2番地と 3番地の掛け算を行って 0番地に書き込むため、dmem.datは、例えば以下のように

しておけば、2× 3が計算されるはずだ。

0000

0001

0002

0003

演習 6

1番地に Xが格納されている。X+(X-1)+...+1を計算するプログラムを書け。

